# Heat transfer project - Halmstad kylteknik AB

Consultants,
Jonathan Eriksson
Albin Svensson



# Problem description

A farmer is considering installing a heat exchanger to cool the milk from the cows before it goes into a storage tank. The milk will be cooled using a plate heat exchanger and enters the primary side of the exchanger at 37 °C. The goal is to cool the milk by at least 10 °C (i.e max 27 °C) before it is transferred to the storage tank for further cooling to 4 °C. To dissipate the heat from the milk, the farmer is considering burying a pipe in the ground, using the ground temperature to cool the pipe and its contents. The pipe would pass through the secondary side of the heat exchanger and would contain a 30% glycol mixture (Propylene glycol).

The milk flows at a rate of 1 liter per minute.
The temperature of the ground is 7 °C.

1. How large of a heat transfer rate should the heat exchanger be able to handle?
2. How long should the pipe in the ground be?
3. At what mass-flow should the glycol mixture flow through the pipe?

A lead that was given:
Place $T_\infty$ = 7 °C at the distance $r_\infty$ from the pipe that corresponds to approximately 2.5 hours of soil heating. Assume the soil is gravel and estimate its thermal conductivity (temperature field) in steady state.

Assumptions:
1. The gravel is dry (worst case)
2. Heat exchanger effectiveness will be at least 0.5.
3. The heat capacity rate is higher on the cold side compared to the low side. The specific heat capacity is almost equal for the hot and cold fluid. Therefore, the assumption implies that the mass flow rate is higher on the cold side. Higher mass flow rate on the cold side is thought to increase heat transfer.

# Solution

## Overview

Heat balance, assuming a value for effectiveness, d-Heat balance and Heat diffusion equation was used to model the problem. This resulted in a calculation that for example can take cross-sectional dimensional parameters as input and give the needed length of the pipe as output. The solution is conservative and serves as a worst case scenario.

1. How large of a heat transfer rate should the heat exchanger be able to handle?
700 W

2. How long should the pipe in the ground be?
70 meters
(Pipe dimensions: Outer diameter = 16mm, wall thickness = 2mm)

3. At what mass-flow should the glycol mixture flow through the pipe?
0.14 kg/s ≈ 8.4 L/min

## CBM




### Heat Balance




Heat balance eq. for inside the heat exchanger gives the heat flow from the hot to the cold side. The heat exchanger needs to be rated for this power. Furthermore, an equation is obtained with $m'_C$, $T_{Cout}$ and $T_{Cin}$ as unknowns.




An assumed value for effectiveness is chosen for the worst case scenario for the effectiveness of the heat exchanger. Most heat exchangers have an effectiveness around 0.75 - 0.9, so it is pretty safe to assume that $\epsilon = 0.5$ is about the worst case possible. $T_{Cin}$ is given by the equation.

## d-Heat Balance







The Nusselt function assumes that the fluid in the pipe is fully developed turbulent flow. Conditions for the equations to be applicable are $R \leq 3 \times 10^3 \wedge 0.5 < \text{Pr} < 2000$.

## Heat diffusion equations







Heat diffusion equation are used in two steps, for the heat flow from the fluid through the plastic and the from the plastic through the gravel until r∞.



# Results and analysis

## Results

### Different diameters

In[ ]:=

| Diameter / Thickness (mm) | 16/2 | 20/2 | 25/2.3 | 32/2 | 32/3 |
|---|---|---|---|---|---|
| Pipe length (m) | 70 | 64 | 59 | 52 | 53 |
| Mass flow rate (kg/s) | 0.14 | 0.17 | 0.21 | 0.26 | 0.25 |

The length of the pipe is a one time cost. A higher mass flow rate of the cold medium will require more power to the pump and thus a higher operational cost. Therefore, we assume that keeping a lower mass flow rate is more important than keeping the pipe length short.
We therefore choose the smallest pipe diameter as this corresponds to the lowest mass flow, i.e 16mm in outer diameter and 2 mm wall thickness (16/2).

### Larger volume flow of milk

We estimate that the cooling system will be able to handle around 10-15 cows during the 2.5 hour period with milk flowing at 1 liter per minute. Below we have scaled the system for double milk capacity.

With milk flowing at 2 liters per minute, and pipe dimensions 16/2:
Pipe length: 132 m
Mass flow rate: 0.14 kg/s
Power: 1400 W

### Achieving lower temperatures

With pipe dimensions 16/2 and cold medium mass flow rate at 0.14 kg/s we can lower the temperature of the milk even further by installing a longer pipe. You can see from the plot below that there is less and less decrease in temperature for each meter of pipe that is added.

### Pipe length vs $T_{Hout}$




# Heat exchanger

## Data

In[ ]:=
```
Remove["Global`*"]
```

In[ ]:=
```
THin = 37; (*°C*)
THout = 27; (*°C*)
T∞ = 7; (*°C*)
cpH = 3890; (*J/kg/K*)
cpC = 3670; (*J/kg/K*)
ρwater = 998; (*kg/m^3*)
ρprop = 1040; (*kg/m^3*)
ρH = 1025; (*kg/m^3*)
ρc = ((ρprop 0.3) + (ρwater 0.7)) // Rationalize;
VH´ = 1/(60 × 10^3); (*m^3/s*)
ϵ = 1/2;
do = 16 × 10^-3;
t = 2 × 10^-3;
di = (do - 2 t) ;
k1 = 4 × 10^-1; (*PE (Medium density)*)
k2 = 4 × 10^-1 ; (*In the upper range for dry gravel (Assumption 1)*)
t1 = 25 × 10^-1 × 60^2; (*s*)
kc = 0.45 // Rationalize; (*W/m/K*)
μc = 325 × 10^-5; (*Pa s*)
νc = μc/ρc;
αc = kc / ρc / cpC;
cpg = 800; (*Conervative estimation for dry gravel (Assumption 1)*)
ρg = 2000;
```

## Mass flow for hot side

In[∘]:=
```
mH´ = ρH VH´;
```

## Heat balance equation

In[∘]:=
```
hbe = Q´ == mH´ cpH (THin - THout) == mC´ cpC (TCout - TCin);
```

### Solution

In[∘]:=
```
{Q´, TCout} = SolveValues[hbe, {Q´, TCout}] // Flatten
```

Out[∘]=

$$\left\{\frac{15\,949}{24}, \frac{15\,949 + 88\,080\,\text{mC}'\,\text{TCin}}{88\,080\,\text{mC}'}\right\}$$

Q is now known. TCout is a function of $T_{\text{Cin}}$ and $m_C$ ´.

## Effectiveness

### Heat capacity rate

Assumption 3 is used here.

In[∘]:=
```
Cmin = mH´ cpH;
Cmax = mC´ cpC;
```

### Maximum heat flow

In[∘]:=
```
Qmax´ = Cmin (THin - TCin);
```

### Effectiveness equation

Here we use assumption 2 in the heat exchanger effectiveness equation.

In[∘]:=

$$\text{eq} = \epsilon == \frac{\text{Q}'}{\text{Qmax}'};$$

### Solution

In[∘]:=
```
{TCin} = SolveValues[eq, TCin];
```

$T_C$ in is now known.

# Gravel

## Base vector

```
In[•]:= ⅈ = {1, 0};
```

## Dummy

$T_1$ and $T_2$ is the temperature in the pipe and the gravel respectively.

```
In[•]:= T₁ = 𝒯1[r];
        T₂ = 𝒯2[r];
```

## Heat flux

```
In[•]:= q´ᵢ_ := -kᵢ Grad[Tᵢ, {r, θ}, "Polar"]
```

## Heat diffusion equation

```
In[•]:= hdeᵢ_ := -Div[q´ᵢ, {r, θ}, "Polar"] == 0
```

## Boundary condtions

```
In[•]:= bchde = (T₁ == T₂ /. r → do/2) && (T₂ == T∞ /. r → r∞) &&
          (q´₁.ⅈ == -h (T₁ - Tm) /. r → di/2) && (q´₁.ⅈ == q´₂.ⅈ /. r → do/2);
```

## Solution

```
In[•]:= {𝒯1[r_], 𝒯2[r_]} = DSolveValue[hde₁ && hde₂ && bchde, {T₁, T₂}, r] // Simplify;
```

# Pipe

## Bulk temperature

In[∘]:= $Tb = \frac{TCin + TCout}{2};$

## Dummy

The mean fluid temperature in the pipe at a distance *x*.

In[∘]:= $Tm = \mathcal{T}m[x];$

## Cross sectional area

In[∘]:= $Acs = \frac{\pi\, di^2}{4};$

## Mean velocity

In[∘]:= $u = \frac{mC'}{\rho c\, Acs};$

## Dimensionless numbers

### Reynolds

In[∘]:= $R = \frac{u\, di}{\nu c};$

### Prandtl

In[∘]:= $Pr = \frac{\nu c}{\alpha c};$

### Nusselt

Formula assumes fully developed turbulent flow. Conditions are $R \geq 3 \times 10^3 \wedge 0.5 < Pr < 2000$.

In[∘]:= $Nu = \frac{(f/8)\ (R - 1000)\ Pr}{1 + \left(127 \times 10^{-1}\right)\ (f/8)^{1/2}\ \left(Pr^{2/3} - 1\right)};$

## Friction factor

Condition is $3\times 10^3 < R < 5\times 10^6$

```
In[∘]:= f = ((0.790 // Rationalize) Log[R] - (1.64 // Rationalize))^-2;
```

## Heat transfer coefficient

```
In[∘]:= h = Nu kc / di;
```

## Heat flux

```
In[∘]:= qs´ = h (T1 - Tm) /. r → di/2;
```

## Heat Flow

```
In[∘]:= dQ´ = qs´ di π;
```

## Enthalpy

```
In[∘]:= H = mC´ cpC Tm;
```

## Enthalpy flow

```
In[∘]:= dH´ = ∂x H;
```

## d-Heat balance

```
In[∘]:= dhb = dQ´ == dH´;
```

## Boundary conditions

```
In[∘]:= bc1 = (Tm == TCout /. x → 0);
```

## d-Heat balance solution

```
In[∘]:= 𝒯m[x_] = DSolveValue[dhb && bc1, Tm, x];
```

## Selection of mass flow cold side

A higher Nusselt number will result in a higher heat transfer coefficient between fluid and pipe.

```
In[•]:= Plot[Nu && R ≥ 3 × 10^3 && 0.5 < Pr < 2000 , {mC´, mH´, 0.4},
  PlotRange → {{0, 0.4}, {0, 130}}, AxesLabel → {"mC´ (kg/s)", "Nu"},
  PlotLabel → "Nusselt number vs mass flow in pipe (turbulent region)"]
```

Out[•]=




Heat transfer will increase as the mass flow increases.

### Choice of mass flow

Having the flow turbulent will increase heat transfer as well as keeping the pipe clean of deposits. Minimizing the mass flow will require a smaller pump as well as less pump power. We therefore choose a mass flow that is low but still in the turbulent region, i.e. minimum value with a added margin of 0.05 kg/s.

```
In[•]:= mC´ = Rationalize[
    ((Minimize[{Nu, R ≥ 3 × 10^3 && 0.5 < Pr < 2000 && Cmax ≥ Cmin}, mC´][[2, 1, 2]]) + 0.05),
    .000001];
```

## Solving for the pipe length

A boundary condition for the end of the pipe is used to solve for the length of the pipe *l*.

```
In[•]:= soll = Solve[Tm == TCin /. x → l, l, Reals] // Flatten // Simplify // Normal;
```

Pipe length *l* is now a function of the distance $r_\infty$.

# Energy volume relation

## Pipe length as a function of r∞

In[ ]:=
```
l = l /. soll;
```

In[ ]:=
```
ℓ[r∞_] = l;
```

## Enthalpy as a function of r∞ and pipe length

In[ ]:=
```
$Assumptions = {r∞ ∈ Reals};
```

The enthalpy in the ground.

In[ ]:=

$$\mathcal{H}[r\infty\_] = \int_0^{\ell[r\infty]} \left( \int_{\frac{do}{2}}^{r\infty} (cpg\,\rho g\,(T_2 - T\infty)\,2\,\pi\,r)\,dr \right) dx\ //\ Normal;$$

The enthalpy is set equal to the power transferred over 2.5 hours.

In[ ]:=
```
solr∞ = FindRoot[Q´ t1 == ℋ[r∞], {r∞, 0.11}, MaxIterations → 10000]
```

Out[ ]=
```
{r∞ → 0.096867}
```

## Pipe length in meters

In[ ]:=
```
l /. solr∞ // N
```

Out[ ]=
```
69.9679
```

# Validation using FEM in ANSYS

## Model

Pipe dimension: Outer diameter 16mm, wall thickness 2mm

Using rotational symmetry to save computational resources, the liquid, pipe and gravel is modelled as as sections of a 10° wedge as shown in the Figure 1, with a length of 0.5 meters in the z-direction. The outer radius of the gravel is set to $r_\infty = 97$ mm.




Figure 1 - Cross section of 10° wedge model

## Simulation approach and validation targets

Steady-state thermal simulations were performed. Because the heat flux and temperature varies along the length (z-direction) of the model, three different simulations are run at $0 \le l \le 0.5\,m$, $35 \le l \le 35.5\,m$ and $69.5 \le l \le 70\,m$, respectively, where $l$ is the length from the inlet of the pipe. The temperature of the fluid at outlet of model (3 data points in total) can then be compared to the calculated values. The simulations will also give heat flux data (6 data points) that is measured at the boundary of the gravel ($r_\infty$) on both inlet side and outlet side of each simulated section. This data can then be compared to calculated values at corresponding locations.

No longer than 0.5 meters of pipe lengths are simulated due to problem size limitations of ANSYS student. Longer pipe section should be simulated in order to potentially discover non-linear behavior of the heat flux along the pipe.

## Materials

All materials assigned to the parts in the simulation have the same material properties as the calculations. Therefore, the simulations can validate that the calculations are correct but they do not validate the underlying (conservative) assumptions about the material properties.

## Boundary conditions

Because the model is 1/36th of a full model, the mass flow needs to be set to $\frac{1}{36} m_C{'} = \frac{0.14}{36} \approx 0.003889\,kg/s$.

Inlet temperature is set to calculated temperature in each of the three simulations, see results for values. Convection heat transfer coefficient if set to the calculated value $h = 2218.6\ W / m^2 / °C$ (Note: Ambient temperature for convection is overridden by the fluid temperature). Outer convex surface of the gravel section is set to 7°C (Conditions C in figure 2). Because of rotational symmetry, an insulated condition is applied on faces in the radial planes of gravel and pipe bodies (dark blue faces in figure 2).




*Figure 2 - Boundary conditions*

## Convergence

Convergence conditions of a minimum of 5% on the maximum heat flux where put on both the pipe and gravel for all simulations. The convergence achieved a difference of 2.97% for the pipe and 0.96% for the gravel in all three simulations, thereby achieving a satisfactory mesh-independent heat flux.

## Results

### Temperature

| l (m) | Calculated fluid temp (°C) | Simulated fluid temp (°C) |
|---|---|---|
| 0 | 18.276 | BC |
| 0.5 | 18.267 | 18.266 |
| 35 | 17.619 | BC |
| 35.5 | 17.610 | 17.610 |
| 69.5 | 17.008 | BC |
| 70 | 16.999 | 16.999 |

*Table 1 - Comparison of calculated fluid temperature at pipe length l.*
*BC - calculated temperature is used as boundary condition for simulation*

### Heat flux

| l (m) | Calculated heat flux $(W/m^2)$ | Simulated heat flux $(W/m^2)$ |
|---|---|---|
| 0 | 16.71 | 16.66 |
| 0.5 | 16.70 | 16.66 |
| 35 | 15.74 | 15.69 |
| 35.5 | 15.72 | 15.69 |
| 69.5 | 14.83 | 14.79 |
| 70 | 14.82 | 14.79 |

## Conclusion

The simulations validate the calculations, showing that the calculations are correct. Therefore, if the assumptions about material properties is correct, a 70 m pipe is the correct length to achieve the desired temperature drop along the pipe.

## Future work

- Investigate how the convection coefficient *h*, changes over the pipe length and if this can have a significant impact.
- Do calculations for a better case where the gravel is damp or saturated to understand how much this improves performance.
- Investigate the recovery time of the gravel. How long time it would take for the heat in the ground to dissipate sufficiently between milking sessions.
- Investigate what pumping power is required to drive the coolant through the system.

## Appendix - Plots

### Section 1

$0 \leq l \leq 0.5\,m$

Temperature plot




Heat flux plot




## Section 2

$35 \leq l \leq 35.5\, m$

Temperature plot



Heat flux plot




## Section 3

$69 \leq l \leq 70\ m$

Temperature plot




Heat flux plot


D: Copy of Copy of Steady-State Thermal
Total Heat Flux 3
Type: Total Heat Flux
Unit: W/m²
Time: 1 s
2026-02-02 11:57:46
178,1 Max
159,96
141,81
123,66
105,51
87,362
69,213
51,065
32,917
14,768 Min
Max
Min
14,788
Node 68994
14,786
Node 68979
0,000
0,100
0,200 (m)
0,050
0,150
Z
Y